WINDOW COORDINATES

PURPOSE
Specifies the region of the graphics device to use for any subsequent graphics or diagrammatic graphics commands.

DESCRIPTION
By default, DATAPLOT uses the entire device when generating graphics and diagrammatic graphics. DATAPLOT defines a 0 to 100 coordinate system in both the horizontal and vertical direction on the device. In certain instances, it is desirable to restrict the output to a specific area of the device. Two common reasons for doing this are:

1. To generate multiple plots per page. Although the MULTIPLOT command can automatically generate multiple plots per page, the WINDOW COORDINATES and FRAME COORDINATES commands provide more control over where an individual plot appears.
2. To generate a plot with a specific aspect ratio. For example, the analyst may want a square plot. Be aware that since DATAPLOT uses a percentage coordinate system, the necessary window coordinates may differ depending on the specific device.

The window is defined by specifying the coordinates for the lower left and upper right corners. A new 0 to 100 coordinate system is defined within this sub-region. Commands based on DATAPLOT’s 0 to 100 coordinate system are then relative to the sub-region, not the full screen.

SYNTAX
WINDOW COORDINATES <x1> <y1> <x2> <y2>
where <x1> is a number or parameter in the range 0 to 100 that specifies the left horizontal coordinate;
<y1> is a number or parameter in the range 0 to 100 that specifies the left vertical coordinate;
<x2> is a number or parameter in the range 0 to 100 that specifies the right horizontal coordinate;
and <y2> is a number or parameter in the range 0 to 100 that specifies the right vertical coordinate;

EXAMPLES
WINDOW COORDINATES 0 0 50 50 --set window to lower left quadrant
WINDOW COORDINATES 0 50 50 100 --set window to upper left quadrant
WINDOW COORDINATES 0 50 100 100 --set window to top half
WINDOW COORDINATES 30 30 70 50 --set window to (30,30) and (70,50)
WINDOW COORDINATES --set window to full screen

NOTE 1
Character sizes and line widths are specified as a percentage of the vertical size of the device (i.e., in DATAPLOT’s 0 to 100 coordinate system). When a window is specified, these values for character sizes and line widths are relative to the sub-region rather than the full device area.

NOTE 2
Coordinates less than 0 or greater than 100 are in fact accepted. However, graphics output that falls outside the 0 to 100 range on the full device is clipped.

NOTE 3
The WINDOW COORDINATES command with no arguments reverts the window coordinates to default.

DEFAULT
The full device area is used (i.e., (0,0) and (100,100)).

SYNONYMS
WINDOW CORNER COORDINATES

RELATED COMMANDS
FRAME CORNER COORDINATES
MULTIPLT
PRE-ERASE

APPLICATIONS
Multi-plotting, presentation graphics
IMPLEMENTATION DATE
Pre-1987

PROGRAM

DIMENSION 100 COLUMNS
SKIP 25; COLUMN LIMITS 20 132
READ AUTO79.DAT Y1 TO Y9
LET N = SIZE Y1
LET CAR = SEQUENCE 1 1 N
LET ZX = DATA 0.5 9.5
LET ZY = DATA 0.5 0.5
XLIMITS 1 9; XTIC OFFSET 0.5 0.5
MAJOR XTIC MARK NUMBER 9; YLIMITS 0 1
TITLE PROFILE PLOT; TITLE SIZE 5
ERASE; PRE-ERASE OFF
.
WINDOW CORNER COORDINATES 0 50 50 100
PROFILE PLOT Y1 Y2 Y3 Y4 Y5 Y6 Y7 Y8 Y9 SUBSET CAR 1
WINDOW CORNER COORDINATES 50 50 100 100
SPIKE ON
PROFILE PLOT Y1 Y2 Y3 Y4 Y5 Y6 Y7 Y8 Y9 SUBSET CAR 1
WINDOW CORNER COORDINATES 0 0 50 50
LINE BLANK; SPIKE BASE 0.5
PROFILE PLOT Y1 Y2 Y3 Y4 Y5 Y6 Y7 Y8 Y9 SUBSET CAR 1
LINE SOLID; SPIKE OFF; PLOT ZY ZX
.
WINDOW CORNER COORDINATES 50 0 100 50
LINE BLANK; SPIKE OFF; BAR ON
BAR BASE 0.5; BAR FILL ON; BAR WIDTH 0.2
PROFILE PLOT Y1 Y2 Y3 Y4 Y5 Y6 Y7 Y8 Y9 SUBSET CAR 1
LINE SOLID; BAR OFF; PLOT ZY ZX